Written in Wolfram Mathematica v11.0

Image processing of clinical trials results

Images are from Figure 1A (intent-to-treat population) of Larkin et al, 2015, Combined Nivolumab and Ipilimumab or Monotherapy in Untreated Melanoma, New England Journal of Medicine; **v373**:p23-34.
Responses to Ipilimumab and Nivolumab were plotted in Larkin et al. with dashed lines; to obtain data within dashes, the figure was first processed in Adobe Illustrator to remove the line dashing option and produce complete solid lines.

Mapping the horizontal and vertical axes

In[1]:=
```
AxesImage = Import[NotebookDirectory[] <> "Larkin combination.png", "PNG"];
```

In[2]:=
```
BinarizedAxesImage = Binarize[AxesImage, 0.7];
MaximumHeight = 2600;
MaximumWidth = 6000;
Show[ImageTake[BinarizedAxesImage, {1, MaximumHeight}, {1, MaximumWidth}],
 ImageSize → {{700}, {700}}]
```

Out[5]=



Vertical axis

Extracting the subset of the image containing the vertical tick marks

In[6]:=
```
(* manually determined positions of the vertical tick marks
 (which columns of pixels contain the vertical axis ticks) *)
VerticalAxisStart = 230;
VerticalAxisEnd = 270;
Show[ImageTake[BinarizedAxesImage, {1, MaximumHeight}, {VerticalAxisStart, VerticalAxisEnd}],
 ImageSize → {{500}, {500}}]
```

Out[8]=

In[9]:=
```
(* plotting the number of colored pixels per row in the subset with vertical tick marks *)
VerticalAxesMarks =
  1 - ImageData[ImageTake[BinarizedAxesImage, {1, MaximumHeight},
     {VerticalAxisStart, VerticalAxisEnd}]];
ListPlot[Map[Total, VerticalAxesMarks], Joined → True]
```

Out[10]=



In[11]:=
```
(* gathering adjacent groups of colored pixels *)
AdjacentVerticalTickMarks =
  Gather[Select[{Range[Length[VerticalAxesMarks]], Map[Total, VerticalAxesMarks]}ᵀ, #[[2]] > 20 &],
   Abs[#1[[1]] - #2[[1]]] < 15 &];
% // TableForm
```

Out[12]//TableForm=

118	119	120	121	122	123	124	125
40	40	40	40	40	40	40	40
312	313	314	315	316	317	318	319
40	40	40	40	40	40	40	40
505	506	507	508	509	510	511	512
40	40	40	40	40	40	40	40
699	700	701	702	703	704	705	706
40	40	40	40	40	40	40	40
893	894	895	896	897	898	899	900
40	40	40	40	40	40	40	40
1087	1088	1089	1090	1091	1092	1093	1094
40	40	40	40	40	40	40	40
1281	1282	1283	1284	1285	1286	1287	1288
40	40	40	40	40	40	40	40
1474	1475	1476	1477	1478	1479	1480	1481
40	40	40	40	40	40	40	40
1668	1669	1670	1671	1672	1673	1674	1675
40	40	40	40	40	40	40	40
1862	1863	1864	1865	1866	1867	1868	1869
40	40	40	40	40	40	40	40
2056	2057	2058	2059	2060	2061	2062	2063
40	40	40	40	40	40	40	40

In[13]:=
```
(* marking the location of each tick mark by the average vertical coordinate *)
VerticalTickPositions =
 {Map[Mean, AdjacentVerticalTickMarks[[1 ;; , All, 1]]], Reverse[(Range[11] - 1) * 10]}ᵀ
```

Out[13]= $\left\{\left\{\frac{243}{2}, 100\right\}, \left\{\frac{631}{2}, 90\right\}, \left\{\frac{1017}{2}, 80\right\}, \left\{\frac{1405}{2}, 70\right\}, \left\{\frac{1793}{2}, 60\right\}, \left\{\frac{2181}{2}, 50\right\}, \left\{\frac{2569}{2}, 40\right\}, \left\{\frac{2955}{2}, 30\right\}, \left\{\frac{3343}{2}, 20\right\}, \left\{\frac{3731}{2}, 10\right\}, \left\{\frac{4119}{2}, 0\right\}\right\}$

In[14]:=
```
(* fitting a straight line to the tick marks positions and the numbers on their labels *)
VerticalTicksFit = FindFit[VerticalTickPositions, m * x + c, {m, c}, x]
```

Out[14]= {m → -0.051602, c → 106.263}

In[15]:=
```
(* comparing the fit (red line) with the positions of the tick marks (black) *)
Show[
 ListPlot[VerticalTickPositions, PlotStyle → Black, Frame → True,
  FrameLabel → {"pixel row", "probability of PFS"}],
 Plot[(m * x + c) /. VerticalTicksFit, {x, 0, MaximumHeight}, PlotStyle → Red]
]
```

Out[15]=



In[16]:=
```
(* position of the zero tick mark (0% PFS) *)
ZeroSurvivalTickPosition = Round[x /. Solve[((m * x + c) /. VerticalTicksFit) == 0, x][[1]]]

(* position of the 100 tick mark (100% PFS) *)
FullSurvivalTickPosition = Round[x /. Solve[((m * x + c) /. VerticalTicksFit) == 100, x][[1]]]
```

Out[16]= 2059

Out[17]= 121

Horizontal axis

Extracting the subset of the image containing the horizontal tick marks

In[18]:=
```
(* manually determined positions of the horizontal tick marks
 (which rows of pixels contain the horizontal axis ticks) *)
HorizontalTickStart = 2095;
HorizontalTickEnd = 2130;

Show[ImageTake[BinarizedAxesImage, {HorizontalTickStart, HorizontalTickEnd}, {1, MaximumWidth}],
 ImageSize → {{1000}, {500}}]
```

Out[20]=

In[21]:=
```
(* plotting the number of colored pixels per column in the subset with horizontal tick marks *)
HorizontalAxesMarks =
  1 -
   ImageData[ImageTake[BinarizedAxesImage, {HorizontalTickStart, HorizontalTickEnd},
      {1, MaximumWidth}]]^T;
ListPlot[Map[Total, HorizontalAxesMarks], Joined → True]
```

Out[22]=



In[23]:=
```
(* gathering adjacent groups of colored pixels *)
AdjacentHorizontalTickMarks =
  Gather[Select[{Range[Length[HorizontalAxesMarks]], Map[Total, HorizontalAxesMarks]}ᵀ,
    #〚2〛 > 10 &], Abs[#1〚1〛 - #2〚1〛] < 20 &];
% // TableForm
```

Out[24]//TableForm=

293	294	295	296	297	298	299	300
19	19	19	19	19	19	19	19
547	548	549	550	551	552	553	554
19	19	19	19	19	19	19	19
802	803	804	805	806	807	808	809
19	19	19	19	19	19	19	19
1056	1057	1058	1059	1060	1061	1062	1063
19	19	19	19	19	19	19	19
1311	1312	1313	1314	1315	1316	1317	1318
19	19	19	19	19	19	19	19
1566	1567	1568	1569	1570	1571	1572	1573
19	19	19	19	19	19	19	19
1820	1821	1822	1823	1824	1825	1826	1827
19	19	19	19	19	19	19	19
2075	2076	2077	2078	2079	2080	2081	2082
19	19	19	19	19	19	19	19
2329	2330	2331	2332	2333	2334	2335	2336
19	19	19	19	19	19	19	19
2584	2585	2586	2587	2588	2589	2590	2591
19	19	19	19	19	19	19	19
2839	2840	2841	2842	2843	2844	2845	2846
19	19	19	19	19	19	19	19
3093	3094	3095	3096	3097	3098	3099	3100
19	19	19	19	19	19	19	19
3348	3349	3350	3351	3352	3353	3354	3355
19	19	19	19	19	19	19	19
3602	3603	3604	3605	3606	3607	3608	3609
19	19	19	19	19	19	19	19
3857	3858	3859	3860	3861	3862	3863	3864
19	19	19	19	19	19	19	19
4112	4113	4114	4115	4116	4117	4118	4119
19	19	19	19	19	19	19	19
4366	4367	4368	4369	4370	4371	4372	4373
19	19	19	19	19	19	19	19
4621	4622	4623	4624	4625	4626	4627	4628
19	19	19	19	19	19	19	19
4875	4876	4877	4878	4879	4880	4881	4882
19	19	19	19	19	19	19	19
5130	5131	5132	5133	5134	5135	5136	5137
19	19	19	19	19	19	19	19
5385	5386	5387	5388	5389	5390	5391	5392
19	19	19	19	19	19	19	19
5639	5640	5641	5642	5643	5644	5645	5646
19	19	19	19	19	19	19	19

In[25]:=
```
(* marking the location of each tick mark by the average horizontal coordinate *)
HorizontalTickPositions =
  {Map[Mean, AdjacentHorizontalTickMarks〚1 ;;, All, 1〛], (Range[22] - 1) * 1}ᵀ
```

Out[25]=

$$\left\{\left\{\frac{593}{2},0\right\},\left\{\frac{1101}{2},1\right\},\left\{\frac{1611}{2},2\right\},\left\{\frac{2119}{2},3\right\},\left\{\frac{2629}{2},4\right\},\left\{\frac{3139}{2},5\right\},\left\{\frac{3647}{2},6\right\},\left\{\frac{4157}{2},7\right\},\left\{\frac{4665}{2},8\right\},\left\{\frac{5175}{2},9\right\},\left\{\frac{5685}{2},10\right\},\left\{\frac{6193}{2},11\right\},\left\{\frac{6703}{2},12\right\},\left\{\frac{7211}{2},13\right\},\left\{\frac{7721}{2},14\right\},\left\{\frac{8231}{2},15\right\},\left\{\frac{8739}{2},16\right\},\left\{\frac{9249}{2},17\right\},\left\{\frac{9757}{2},18\right\},\left\{\frac{10267}{2},19\right\},\left\{\frac{10777}{2},20\right\},\left\{\frac{11285}{2},21\right\}\right\}$$

In[26]:=
```
(* fitting a straight line to the tick marks positions and the numbers on their labels *)
HorizontalTicksFit = FindFit[HorizontalTickPositions, m * x + c, {m, c}, x]
```

Out[26]=

```
{m → 0.00392777, c → -1.16316}
```

In[27]:=
```
(* comparing the fit (red line) with the positions of the tick marks (black) *)
Show[
 ListPlot[HorizontalTickPositions, PlotStyle → Black, Frame → True,
  FrameLabel → {"pixel row", "Time (months)"}],
 Plot[(m * x + c) /. HorizontalTicksFit, {x, 0, MaximumWidth}, PlotStyle → Red]
]
```

Out[27]=



In[28]:=
```
(* position of the Time=0 tick mark *)
TimeZeroTickPosition = Round[x /. Solve[((m * x + c) /. HorizontalTicksFit) == 0, x][[1]]]
```

Out[28]= 296

In[29]:=
```
TimeZeroTickPosition
```

Out[29]= 296

In[30]:=
```
(* this function takes a pixel coordinate (number of pixels to the right of the Time=0 point)
 and converts it to a number of months of PFS *)
SurvivingMonthsFromPixelCoordinate[numberofhorizontalpixels_] :=
 (m /. HorizontalTicksFit) * numberofhorizontalpixels
```

Processing Ipilimumab response

In[31]:=
```
Mono1Image = Import[NotebookDirectory[] <> "Larkin ipilimumab.png", "PNG"];
```

In[32]:=
```
BinarizedMono1Image = Binarize[Mono1Image, 0.7];
Show[ImageTake[BinarizedMono1Image, {1, MaximumHeight}, {1, MaximumWidth}],
 ImageSize → {{700}, {700}}]
```

Out[33]=



```
In[34]:= (* using the locations of 'zero' tick marks on horizontal and vertical axes to focus
          on the relevant portion of the plot *)
         Mono1MappingImagePart =
          Show[ImageTake[BinarizedMono1Image, {FullSurvivalTickPosition, ZeroSurvivalTickPosition},
            {TimeZeroTickPosition, MaximumWidth}], ImageSize → {{700}, {700}}]
```

Out[34]=



```
In[35]:= Mono1MappingImageData = 1 - ImageData[Mono1MappingImagePart];
```

```
In[36]:= (* how many pixels are in each row between the axis and the survival function? *)
         RunsOfConsecutiveElementsPerRow =
           Table[{i, Map[Length, Split[Mono1MappingImageData⟦i⟧, Chop[#2 - #1] == 0 &]]},
            {i, 1, ZeroSurvivalTickPosition - FullSurvivalTickPosition}];
         SurvivalPixelsPerRow =
           Map[{#⟦1⟧, #⟦2, 2⟧} &, Select[RunsOfConsecutiveElementsPerRow, Length[#⟦2⟧] > 2 &]];
         SurvivingMonthsPerPixelRow =
           Map[{SurvivingMonthsFromPixelCoordinate[#⟦2⟧],
              (ZeroSurvivalTickPosition - FullSurvivalTickPosition) - #⟦1⟧} &, SurvivalPixelsPerRow];
```

```
In[39]:= (* capping 'survival' time at 17 months (the effective end of the data) *)
         MaximumSurvivalMonths = 17;
         (* building a survival distribution from counts of pixels in each row *)
         Mono1SurvivalDistribution = Table[{i,
             If[
              (* test *)Select[SurvivingMonthsPerPixelRow, #⟦2⟧ == i &] == {},
              (* true; no progression time found *)MaximumSurvivalMonths,
              (* false; use observed progression time *)
              Select[SurvivingMonthsPerPixelRow, #⟦2⟧ == i &]⟦1, 1⟧
             ]
            }, {i, 1, ZeroSurvivalTickPosition - FullSurvivalTickPosition}];
```

In[41]:=
```
(* plotting the resulting survival function *)
Plot[SurvivalFunction[EmpiricalDistribution[Mono1SurvivalDistribution[[All, 2]]]][x],
 {x, 0, 16}, Exclusions → None, ImageSize → {{500}, {500}}, Frame → {{True, False}, {True, False}},
 FrameStyle → Directive[Black, Thickness[Medium]],
 BaseStyle → {FontFamily → "Arial", FontSize → 12}, PlotRangePadding → None,
 PlotRange → {{0, 15}, {0, 1}}, FrameLabel → {"Progression free survival (months)", "Probability"}]
```

Out[41]=



Processing Nivolumab response

In[42]:=
```
Mono2Image = Import[NotebookDirectory[] <> "Larkin nivolumab.png", "PNG"];
```

In[43]:=
```
BinarizedMono2Image = Binarize[Mono2Image, 0.7];
Show[ImageTake[BinarizedMono2Image, {1, MaximumHeight}, {1, MaximumWidth}],
 ImageSize → {{700}, {700}}]
```

Out[44]=



In[45]:=
```
(* using the locations of 'zero' tick marks on horizontal and vertical axes to focus
 on the relevant portion of the plot *)
Mono2MappingImagePart =
 Show[ImageTake[BinarizedMono2Image, {FullSurvivalTickPosition, ZeroSurvivalTickPosition},
   {TimeZeroTickPosition, MaximumWidth}], ImageSize → {{700}, {700}}]
```

Out[45]=



In[46]:=
```
Mono2MappingImageData = 1 - ImageData[Mono2MappingImagePart];
```

In[47]:=
```
(* how many pixels are in each row between the axis and the survival function? *)
RunsOfConsecutiveElementsPerRow =
  Table[{i, Map[Length, Split[Mono2MappingImageData⟦i⟧, Chop[#2 - #1] == 0 &]]},
   {i, 1, ZeroSurvivalTickPosition - FullSurvivalTickPosition}];
SurvivalPixelsPerRow =
  Map[{#⟦1⟧, #⟦2, 2⟧} &, Select[RunsOfConsecutiveElementsPerRow, Length[#⟦2⟧] > 2 &]];
SurvivingMonthsPerPixelRow =
  Map[{SurvivingMonthsFromPixelCoordinate[#⟦2⟧],
     (ZeroSurvivalTickPosition - FullSurvivalTickPosition) - #⟦1⟧} &, SurvivalPixelsPerRow];
```

In[50]:=
```
(* capping 'survival' time at 17 months (the effective end of the data) *)
MaximumSurvivalMonths = 17;
(* building a survival distribution from counts of pixels in each row *)
Mono2SurvivalDistribution = Table[{i,
     If[
      (* test *)Select[SurvivingMonthsPerPixelRow, #⟦2⟧ == i &] == {},
      (* true; no progression time found *)MaximumSurvivalMonths,
      (* false; use observed progression time *)
      Select[SurvivingMonthsPerPixelRow, #⟦2⟧ == i &]⟦1, 1⟧
     ]
    }, {i, 1, ZeroSurvivalTickPosition - FullSurvivalTickPosition}];
```

```
In[52]:= (* plotting the resulting survival function *)
Plot[SurvivalFunction[EmpiricalDistribution[Mono2SurvivalDistribution[[All, 2]]]][x],
 {x, 0, 16}, Exclusions → None, ImageSize → {{500}, {500}}, Frame → {{True, False}, {True, False}},
 FrameStyle → Directive[Black, Thickness[Medium]],
 BaseStyle → {FontFamily → "Arial", FontSize → 12}, PlotRangePadding → None,
 PlotRange → {{0, 15}, {0, 1}}, FrameLabel → {"Progression free survival (months)", "Probability"}]
```

Out[52]=



Processing Combination response

```
In[53]:= ComboImage = Import[NotebookDirectory[] <> "Larkin combination.png", "PNG"];

In[54]:= BinarizedComboImage = Binarize[ComboImage, 0.7];
Show[ImageTake[BinarizedComboImage, {1, MaximumHeight}, {1, MaximumWidth}],
 ImageSize → {{700}, {700}}]
```

Out[55]=



In[56]:=
```
(* using the locations of 'zero' tick marks on horizontal and vertical axes to focus
 on the relevant portion of the plot *)
ComboMappingImagePart =
 Show[ImageTake[BinarizedComboImage, {FullSurvivalTickPosition, ZeroSurvivalTickPosition},
   {TimeZeroTickPosition, MaximumWidth}], ImageSize → {{700}, {700}}]
```

Out[56]=



In[57]:=
```
ComboMappingImageData = 1 - ImageData[ComboMappingImagePart];
```

In[58]:=
```
(* how many pixels are in each row between the axis and the survival function? *)
RunsOfConsecutiveElementsPerRow =
  Table[{i, Map[Length, Split[ComboMappingImageData[[i]], Chop[#2 - #1] == 0 &]]},
   {i, 1, ZeroSurvivalTickPosition - FullSurvivalTickPosition}];
SurvivalPixelsPerRow =
  Map[{#[[1]], #[[2, 2]]} &, Select[RunsOfConsecutiveElementsPerRow, Length[#[[2]]] > 2 &]];
SurvivingMonthsPerPixelRow =
  Map[{SurvivingMonthsFromPixelCoordinate[#[[2]]],
     (ZeroSurvivalTickPosition - FullSurvivalTickPosition) - #[[1]]} &, SurvivalPixelsPerRow];
```

In[61]:=
```
(* capping 'survival' time at 17 months (the effective end of the data) *)
MaximumSurvivalMonths = 17;
(* building a survival distribution from counts of pixels in each row *)
ComboSurvivalDistribution = Table[{i,
     If[
      (* test *)Select[SurvivingMonthsPerPixelRow, #[[2]] == i &] == {},
      (* true; no progression time found *)MaximumSurvivalMonths,
      (* false; use observed progression time *)
      Select[SurvivingMonthsPerPixelRow, #[[2]] == i &][[1, 1]]
     ]
    }, {i, 1, ZeroSurvivalTickPosition - FullSurvivalTickPosition}];
```

In[63]:=
```
(* plotting the resulting survival function *)
Plot[SurvivalFunction[EmpiricalDistribution[ComboSurvivalDistribution[[All, 2]]]][x],
 {x, 0, 16}, Exclusions → None, ImageSize → {{500}, {500}}, Frame → {{True, False}, {True, False}},
 FrameStyle → Directive[Black, Thickness[Medium]],
 BaseStyle → {FontFamily → "Arial", FontSize → 12}, PlotRangePadding → None,
 PlotRange → {{0, 15}, {0, 1}}, FrameLabel → {"Progression free survival (months)", "Probability"}]
```

Out[63]=



Comparing observed combination with prediction from independent drug action

Randomly sampling monotherapy response distributions with no correlation

In[64]:=
```
(* Simulate 5000 patients receiving combination
 therapy: each receives a randomly sampled response from each monotherapy response distribution,
and the patients duration of PFS is the longer of the two sampled responses
 (the principle of independent drug action) *)
UncorrelatedSamplesOfMonotherapyResponses =
  Table[Max[{RandomChoice[Mono1SurvivalDistribution[[All, 2]]],
     RandomChoice[Mono2SurvivalDistribution[[All, 2]]]}], {5000}];
```

Randomly sampling monotherapy response distributions with partial correlation

This is achieved by starting with the two monotherapies response distributions in a perfectly correlated joint distribution, and then adding a random number (up to a user-specified size) to the rank (that is, row number) of each entry. As the degree of rank-randomization increases, the rank correlation decreases from 1 and approaches 0. Identifying the appropriate value of the rank randomization parameter to produce a desired Spearman Rank Correlation is matter of manual adjustment.

In[65]:=
```
(* how many repeats of the complete data set to merge together
  (for purpose of minimizing variance between individual repeats of the 'rank
    randomization' process *)
resamplingfactor = 1; (* '1' is sufficient for this data set;
but a larger value may be helpful for data sets derived from lower-resolution images *)
(* this is the amount of rank randomization that produces Spearman Rank Correlation
 of approximately 0.25 *)
amountofrankrandomization = 1800;
ReSampledMono1Distribution =
  Sort[Flatten[Table[Mono1SurvivalDistribution〚All, 2〛, {resamplingfactor}]]];
ReSampledMono2Distribution =
  Sort[Flatten[Table[Mono2SurvivalDistribution〚All, 2〛, {resamplingfactor}]]];

(* response distributions in partially randomized order *)
SlightlyRandomizedMONO1Distribution =
  Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
     ReSampledMono1Distribution〚i〛}, {i, 1, Length[ReSampledMono1Distribution]}],
   #1〚1〛 < #2〚1〛 &];
SlightlyRandomizedMONO2Distribution =
  Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
     ReSampledMono2Distribution〚i〛}, {i, 1, Length[ReSampledMono2Distribution]}],
   #1〚1〛 < #2〚1〛 &];
(* computing the Spearman Rank Correlation of the partially randomized response
 distributions *)
SpearmanRho[SlightlyRandomizedMONO1Distribution〚All, 2〛,
 SlightlyRandomizedMONO2Distribution〚All, 2〛]
(* applying the principle of independent drug action: in each row (representing a patient),
the drug response (duration of PFS) is taken to be the best one of the two sampled
 monotherapy responses *)
PartiallyCorrelatedSamplesOfMonotherapyResponses =
  Table[Max[{SlightlyRandomizedMONO1Distribution〚i, 2〛,
     SlightlyRandomizedMONO2Distribution〚i, 2〛}], {i, 1, Length[ReSampledMono1Distribution]}];
```

Out[71]= 0.284383

In[73]:=
```
(* Repeating the randomization procedure 100 times to show that response correlation
 is tightly distributed around the targeted value of 0.25 *)
TableOfCorrelationsFromRepeats = Table[
   amountofrankrandomization = 1800;
   SlightlyRandomizedMONO1Distribution =
    Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
       ReSampledMono1Distribution[[i]]}, {i, 1, Length[ReSampledMono1Distribution]}],
     #1[[1]] < #2[[1]] &];
   SlightlyRandomizedMONO2Distribution =
    Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
       ReSampledMono2Distribution[[i]]}, {i, 1, Length[ReSampledMono2Distribution]}],
     #1[[1]] < #2[[1]] &];

   SpearmanRho[SlightlyRandomizedMONO1Distribution[[All, 2]],
    SlightlyRandomizedMONO2Distribution[[All, 2]]]
   , {100}];
```

In[74]:=
```
Print["Mean response correlation = " <>
  ToString[Round[Mean[TableOfCorrelationsFromRepeats], 0.001]]]
Histogram[TableOfCorrelationsFromRepeats, {0., 1.0, 0.01}, "Probability", Frame → True,
 FrameLabel → {"Spearman Rank Correlation", "Probability"}]
```

Mean response correlation = 0.264

Out[75]=


In[76]:=
```
(* computing median PFS for each condition: the observed monotherapies,
the observed combination, and the simulated effects with no response correlation or
 with partial response correlation *)
Mono1Median = Median[Mono1SurvivalDistribution[[All, 2]]]
Mono2Median = Median[Mono2SurvivalDistribution[[All, 2]]]
ComboMedian = Median[ComboSurvivalDistribution[[All, 2]]]
Sim1Median = Median[UncorrelatedSamplesOfMonotherapyResponses]
Sim2Median = Median[PartiallyCorrelatedSamplesOfMonotherapyResponses]
```

Out[76]= 2.9851

Out[77]= 7.10141

Out[78]= 12.8949

Out[79]= 14.2814

Out[80]= 12.6867

```
In[81]:= SurvivalPlot = Plot[{
    SurvivalFunction[EmpiricalDistribution[Mono1SurvivalDistribution[[All, 2]]]][x],
    SurvivalFunction[EmpiricalDistribution[Mono2SurvivalDistribution[[All, 2]]]][x],
    SurvivalFunction[EmpiricalDistribution[ComboSurvivalDistribution[[All, 2]]]][x],
    SurvivalFunction[EmpiricalDistribution[PartiallyCorrelatedSamplesOfMonotherapyResponses]][
      x],
    SurvivalFunction[EmpiricalDistribution[UncorrelatedSamplesOfMonotherapyResponses]][x]
   }
   , {x, 0, 16}, PlotRange → {{0, 15}, {0, 1.11}}, Exclusions → None,
   PlotStyle → {Directive[RGBColor[0.8, 0.1, 0.6], AbsoluteThickness[3]],
     Directive[ColorData[3, 4], AbsoluteThickness[3]],
     Directive[ColorData[3, 6], AbsoluteThickness[3]],
     Directive[Black, Dashing[{0.2, 0.2}] AbsoluteThickness[2]],
     Directive[GrayLevel[0.5], AbsoluteThickness[2]]},
   BaseStyle → {FontFamily → "Arial", FontSize → 12},
   FrameStyle → Directive[Black, Thickness[Medium]], Frame → {{True, False}, {True, False}},
   FrameTicks → {{Table[{i, 100 * i, {0, 0.015}}, {i, 0, 1, 1 / 5}], None},
     {Table[{i, i, {0, 0.015}}, {i, 0, 60, 3}], None}},
   FrameLabel → {Style["Progression free survival (months)"], Style["Patients (%)"]},
   Prolog → {Gray, Thickness[Medium], Lighter[Yellow, 0.75], EdgeForm[None],
     Rectangle[{0, 1}, {15, 1.11}],
     Black, Text["Median:", {0.2, 1.03}, {-1, -1}], RGBColor[0.8, 0.1, 0.6],
     Text[ToString[NumberForm[Mono1Median, {3, 1}]], {Mono1Median, 1.03}, {0, -1}],
     Darker[ColorData[3, 4], 0.1], Text[ToString[NumberForm[Mono2Median, {3, 1}]],
       {Mono2Median, 1.03}, {0, -1}], ColorData[3, 6],
     Text[ToString[NumberForm[ComboMedian, {3, 1}]], {ComboMedian, 1.03}, {0, -1}],
     GrayLevel[0.5], Text[ToString[NumberForm[Sim1Median, {3, 1}]], {Sim1Median, 1.0}, {0, 1}],
     Black, Text[ToString[NumberForm[Sim2Median, {3, 1}]], {Sim2Median, 1.0}, {0, 1}],
     AbsoluteThickness[2], RGBColor[0.8, 0.1, 0.6], Line[{{Mono1Median, 1}, {Mono1Median, 1.03}}],
     Darker[ColorData[3, 4], 0.1], Line[{{Mono2Median, 1}, {Mono2Median, 1.03}}],
     GrayLevel[0.5], Opacity[1], Line[{{Sim1Median, 1}, {Sim1Median, 1.03}}], Black,
     Line[{{Sim2Median, 1}, {Sim2Median, 1.03}}], ColorData[3, 6],
     Line[{{ComboMedian, 1}, {ComboMedian, 1.03}}]},
   AspectRatio → 1 / 2, Filling → None, FillingStyle -> Directive[Opacity[1], GrayLevel[0.75]],
   ImageSize → {{1000}, {250}}, ImagePadding → {{60, 10}, {60, 10}}, PlotPoints → 500,
   Epilog -> {EdgeForm[Directive[Black, Thickness[Medium]]], White, Opacity[0.7],
     Opacity[1], EdgeForm[None], CapForm["Butt"], AbsoluteThickness[3], AbsolutePointSize[8],
     RGBColor[0.8, 0.1, 0.6], Line[{{0.4, 0.38}, {1, 0.38}}], ColorData[3, 4],
     Line[{{0.4, 0.3}, {1, 0.3}}], ColorData[3, 6], Line[{{0.4, 0.22}, {1, 0.22}}],
     GrayLevel[0.5], Opacity[1], AbsoluteThickness[2], Line[{{0.4, 0.14}, {1, 0.14}}],
     Black, Line[{{0.4, 0.06}, {1, 0.06}}], Black, FontFamily → "Arial", FontSize → 11,
     Text[Style["Ipilimumab", FontSize → 11], {1.15, 0.38}, {-1, 0}],
     Text[Style["Nivolumab", FontSize → 11], {1.15, 0.3}, {-1, 0}],
     Text[Style["Combination", FontSize → 11], {1.15, 0.22}, {-1, 0}],
     Text[Style["Random sampling (ρ=0)", FontSize → 11], {1.15, 0.14}, {-1, 0}],
     Text[Style["Correlated sampling (ρ=0.25)", FontSize → 11], {1.15, 0.06}, {-1, 0}]}]
```

Out[81]=


Median: 3.0 7.1 12.9
12.7 14.3
Patients (%)
100
80
60
40
20
0
Ipilimumab
Nivolumab
Combination
Random sampling (ρ=0)
Correlated sampling (ρ=0.25)
0 3 6 9 12 15
Progression free survival (months)

In[82]:=
```
(*vertical offset of median bar *)
vo = -0.985;

SurvivalPlot = Plot[{
   SurvivalFunction[EmpiricalDistribution[Mono1SurvivalDistribution[[All, 2]]]][x],
   SurvivalFunction[EmpiricalDistribution[Mono2SurvivalDistribution[[All, 2]]]][x],
   SurvivalFunction[EmpiricalDistribution[ComboSurvivalDistribution[[All, 2]]]][x],
   SurvivalFunction[EmpiricalDistribution[PartiallyCorrelatedSamplesOfMonotherapyResponses]][
    x],
   SurvivalFunction[EmpiricalDistribution[UncorrelatedSamplesOfMonotherapyResponses]][x]
  }
  , {x, 0, 16}, PlotRange → {{0, 15}, {0, 1.11}}, Exclusions → None,
  PlotStyle → {Directive[RGBColor[0.8, 0.1, 0.6], AbsoluteThickness[3]],
    Directive[ColorData[3, 4], AbsoluteThickness[3]],
    Directive[ColorData[3, 6], AbsoluteThickness[3]],
    Directive[Black, Dashing[{0.2, 0.2}] AbsoluteThickness[2]],
    Directive[GrayLevel[0.5], AbsoluteThickness[2]]},
  BaseStyle → {FontFamily → "Arial", FontSize → 12},
  FrameStyle → Directive[Black, Thickness[Medium]], Frame → {{True, False}, {True, False}},
  FrameTicks → {{Table[{i, 100 * i, {0, 0.015}}, {i, 0, 1, 1 / 5}], None},
    {Join[Table[{i, i, {0, 0.015}}, {i, 0, 15, 3}], Table[{i, , {0, 0.01}}, {i, 0, 15, 1}]], None}},
  FrameLabel → {Style["Progression free survival (months)"], Style["Patients (%)"]},
  Prolog → {Gray, Thickness[Medium], Lighter[Yellow, 0.75], EdgeForm[None],
    Rectangle[{0, 1 + vo - 0.02}, {15, 1.11 + vo}],
    Black, Text["Median:", {0.2, 1.03 + vo}, {-1, -1}], RGBColor[0.8, 0.1, 0.6],
    Text[ToString[NumberForm[Mono1Median, {3, 1}]], {Mono1Median, 1.03 + vo}, {0, +vo}],
    Darker[ColorData[3, 4], 0.1], Text[ToString[NumberForm[Mono2Median, {3, 1}]],
     {Mono2Median, 1.03 + vo}, {0, -1}], ColorData[3, 6],
    Text[ToString[NumberForm[ComboMedian, {3, 1}]], {ComboMedian, 1.03 + vo}, {-0.9, -1}],
    GrayLevel[0.5], Text[ToString[NumberForm[Sim1Median, {3, 1}]], {Sim1Median, 1.03 + vo},
     {-0.4, -1}], Black, Text[ToString[NumberForm[Sim2Median, {3, 1}]], {Sim2Median, 1.03 + vo},
     {1, -1}], AbsoluteThickness[2], RGBColor[0.8, 0.1, 0.6],
    Line[{{Mono1Median, 1 + vo - 0.015}, {Mono1Median, 1.03 + vo}}], Darker[ColorData[3, 4], 0.1],
    Line[{{Mono2Median, 1 + vo - 0.015}, {Mono2Median, 1.03 + vo}}], GrayLevel[0.5],
    Opacity[1], Line[{{Sim1Median, 1 + vo - 0.015}, {Sim1Median, 1.03 + vo}}], Black,
    CapForm["Round"], Line[{{Sim2Median, 1 + vo - 0.015}, {Sim2Median, 1.02 + vo}}],
    Line[{{Sim2Median, 1.02 + vo}, {Sim2Median - 0.15, 1.03 + vo}}], ColorData[3, 6],
    Line[{{ComboMedian, 1 + vo - 0.015}, {ComboMedian, 1.02 + vo}}],
    Line[{{ComboMedian, 1.02 + vo}, {ComboMedian + 0.15, 1.03 + vo}}]},
  AspectRatio → 1 / 2, Filling → None, FillingStyle -> Directive[Opacity[1], GrayLevel[0.75]],
  ImageSize → {{1000}, {250}}, ImagePadding → {{60, 10}, {60, 10}}, PlotPoints → 500]

Export[NotebookDirectory[] <> "Figure 1B, ipilimumab plus nivolumab PFS.pdf",
 SurvivalPlot, "PDF"]
```

Out[83]=



Plotting the dependence of predicted PFS on correlation

In[85]:=
```
(* this function takes as its input an amount of rank randomization,
and returns a response correlation and predicted median PFS at this level of
 correlation in drug response *)
PFSvsCorrelation[amountofrankrandomization_] := Module[{},

  SlightlyRandomizedMONO1Distribution =
   Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
      Mono1SurvivalDistribution[[i, 2]]}, {i, 1, Length[Mono1SurvivalDistribution]}],
    #1[[1]] < #2[[1]] &];
  SlightlyRandomizedMONO2Distribution =
   Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
      Mono2SurvivalDistribution[[i, 2]]}, {i, 1, Length[Mono2SurvivalDistribution]}],
    #1[[1]] < #2[[1]] &];

  ResponseCorrelation = SpearmanRho[SlightlyRandomizedMONO1Distribution[[All, 2]],
    SlightlyRandomizedMONO2Distribution[[All, 2]]];
  PartiallyCorrelatedSamplesOfMonotherapyResponses =
   Table[Max[{SlightlyRandomizedMONO1Distribution[[i, 2]],
      SlightlyRandomizedMONO2Distribution[[i, 2]]}], {i, 1, Length[Mono1SurvivalDistribution]}];

  MedianPFS = Median[PartiallyCorrelatedSamplesOfMonotherapyResponses];

  {ResponseCorrelation, MedianPFS}
 ]
```

In[86]:=
```
(* this function takes as its input an amount of rank randomization,
and returns a response correlation and predicted median PFS at this level of
 correlation in drug response *)
PFSvsNegativeCorrelation[amountofrankrandomization_] := Module[{},

  SlightlyRandomizedMONO1Distribution =
   Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
      Mono1SurvivalDistribution[[i, 2]]}, {i, 1, Length[Mono1SurvivalDistribution]}],
    #1[[1]] < #2[[1]] &];
  SlightlyRandomizedMONO2Distribution =
   Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
      Reverse[Mono2SurvivalDistribution][[i, 2]]}, {i, 1, Length[Mono2SurvivalDistribution]}],
    #1[[1]] < #2[[1]] &];

  ResponseCorrelation = SpearmanRho[SlightlyRandomizedMONO1Distribution[[All, 2]],
    SlightlyRandomizedMONO2Distribution[[All, 2]]];
  PartiallyCorrelatedSamplesOfMonotherapyResponses =
   Table[Max[{SlightlyRandomizedMONO1Distribution[[i, 2]],
      SlightlyRandomizedMONO2Distribution[[i, 2]]}], {i, 1, Length[Mono1SurvivalDistribution]}];

  MedianPFS = Median[PartiallyCorrelatedSamplesOfMonotherapyResponses];

  {ResponseCorrelation, MedianPFS}
 ]
```

In[87]:=
```
(* executing the above function over a wide range of rank randomization values
 (log-distributed to achieve the necessary range) *)
manycorrelations = Table[PFSvsCorrelation[10^i], {i, 0., 5.5, 0.01}];
```

In[88]:=
```
(* executing the above function over a wide range of rank randomization values
  (log-distributed to achieve the necessary range) *)
manynegativecorrelations = Table[PFSvsNegativeCorrelation[10^i], {i, 0., 5.5, 0.01}];
```

In[89]:=
```
joinedcorrelations = Join[manycorrelations, manynegativecorrelations];
```

In[90]:=
```
(* fitting a linear-plus-sigmoidal curve to the relationship between response
  correlation and PFS *)
nh = 2;
```

$$\text{model} = b - h * \frac{(x+1)^{nh}}{(x+1)^{nh} + k^{nh}} - m * x;$$

```
sigmoidfit50 = NonlinearModelFit[joinedcorrelations, model, {{b, 17}, {h, 5}, {k, 1}, {m, 8}}, x]

Show[
 ListPlot[joinedcorrelations, PlotRange → {0, 20}, PlotStyle → Red],
 Plot[sigmoidfit50[x], {x, -1, 1}, PlotRange → All, PlotStyle → Gray],
 Frame → True, FrameLabel → {"Response correlation", "Median PFS"}]
```

Out[92]= FittedModel[$20.1022 + 3.30831\,x - \frac{49.2589\,(1+x)^2}{7.91918 + (1+x)^2}$]

Out[93]=



In[94]:=
```
Mono1Median = Quantile[Mono1SurvivalDistribution[[All, 2]], 0.5]
Mono2Median = Quantile[Mono2SurvivalDistribution[[All, 2]], 0.5]
ComboMedian = Quantile[ComboSurvivalDistribution[[All, 2]], 0.5]
```

Out[94]= 2.9851

Out[95]= 7.10141

Out[96]= 12.8949

```
In[97]:= PFSvsCorrelationPlot = Plot[{sigmoidfit50[x], ComboMedian, Mono2Median, Mono1Median},
  {x, 0, 1},
  PlotStyle → {Directive[Black, AbsoluteThickness[2]],
    Directive[Dashing[{0.05, 0.05}], ColorData[3, 6], AbsoluteThickness[2]],
    Directive[Dashing[{0.05, 0.05}], ColorData[3, 4], AbsoluteThickness[2]],
    Directive[Dashing[{0.05, 0.05}], RGBColor[0.8, 0.1, 0.6], AbsoluteThickness[2]]},
  Filling → None, FillingStyle → Directive[ColorData[3, 6], Opacity[0.3]],
  PlotRange → {{-0.03, 1.0}, {0, 17}}, Frame → {{True, True}, {True, False}},
  FrameStyle → {{Directive[Black, Dashing[{1.01, 1}], Thickness[Medium]],
     Directive[White, Opacity[0]]}, {Directive[Black, Thickness[Medium]],
     Directive[Black, Thickness[Medium]]}}, Axes → False,
  BaseStyle → {FontFamily → "Arial", FontSize → 12}, AspectRatio → 1 * 17 / 15,
  ImageSize → {{1000}, {200}}, ImagePadding → {{50, 40}, {50, 10}},
  FrameTicks →
   {{Table[{i, i, {0, 0.035}}, {i, 0, 20, 3}],
     {{Mono1Median, Style[ToString[NumberForm[Mono1Median, {3, 1}]], RGBColor[0.8, 0.1, 0.6],
        Opacity[1]], {0, 0}},
      {Mono2Median, Style[ToString[NumberForm[Mono2Median, {3, 1}]], ColorData[3, 4], Opacity[1]],
       {0, 0}}, {ComboMedian, Style[ToString[NumberForm[ComboMedian, {3, 1}]],
        ColorData[3, 6], Opacity[1]], {0, 0}}}},
    {Join[Table[{N[i], i, {0, 0.04}}, {i, -1, 1, 1 / 2}],
      Table[{N[i], , {0, 0.02}}, {i, -1, 1, 1 / 10}]] /. {1 / 2 → "0.5", -1 / 2 → "-0.5"}, None}},
  FrameLabel → {"Correlation in response, ρ ", "Median PFS (months)"},
  Epilog → {GrayLevel[0.5], AbsolutePointSize[7], Point[{0, sigmoidfit50[0]}], Black,
    Point[{0.25, sigmoidfit50[0.25]}], Dashing[{0.001, 0.025}], CapForm["Round"],
    AbsoluteThickness[1.5], Line[{{0.25, sigmoidfit50[0.25]}, {0.25, 0}}],
    Text[ToString[NumberForm[Sim2Median, {3, 1}]], {0.25, sigmoidfit50[0.25] + 0.2}, {-0.9, -1}],
    GrayLevel[0.5], Text[ToString[NumberForm[Sim1Median, {3, 1}]], {0., sigmoidfit50[0.] + 0.2},
     {-0.9, -1}]}]

Export[NotebookDirectory[] <> "Figure 1C, Median PFS vs Correlation.pdf",
 PFSvsCorrelationPlot, "PDF"]
```

Out[97]=



In[99]:=
```
(* supplementary figure: relationship between median PFS and correlation,
showing each simulation point *)
Show[
 Plot[{sigmoidfit50[x], ComboMedian, Mono2Median, Mono1Median}, {x, 0, 1},
  PlotStyle → {Directive[GrayLevel[0.25], AbsoluteThickness[2]],
    Directive[Dashing[{0.04, 0.05}], ColorData[3, 6], AbsoluteThickness[2]],
    Directive[Dashing[{0.04, 0.05}], ColorData[3, 4], AbsoluteThickness[2]],
    Directive[Dashing[{0.04, 0.05}], RGBColor[0.8, 0.1, 0.6], AbsoluteThickness[2]]},
  Filling → None, FillingStyle → Directive[ColorData[3, 6], Opacity[0.3]],
  PlotRange → {{-0.03, 1.0}, {0, 15}}, Frame → {{True, False}, {True, False}},
  FrameStyle → Directive[Black, Thickness[Medium]], Axes → False,
  BaseStyle → {FontFamily → "Arial", FontSize → 12}, AspectRatio → 1, ImageSize → {{1000}, {170}},
  ImagePadding → {{45, 10}, {45, 10}},
  FrameTicks →
   {Join[Table[{N[i], i, {0, 0.04}}, {i, -1, 1, 1 / 2}],
      Table[{N[i], , {0, 0.025}}, {i, -1, 1, 1 / 10}]] /. {1 / 2 → "0.5", -1 / 2 → "-0.5"},
    Table[{i, i, {0, 0.035}}, {i, 0, 20, 3}]},
  FrameLabel → {"Correlation in response ", "Median PFS (months)"}]
 ,
 ListPlot[manycorrelations, PlotRange → {{-1, 1}, {0, 18}},
  Epilog → {Dashing[{0.05, 0.03}], ColorData[3, 6], AbsoluteThickness[2],
    Line[{{-1, ComboMedian}, {1, ComboMedian}}], ColorData[3, 4],
    Line[{{-1, Mono2Median}, {1, Mono2Median}}], RGBColor[0.8, 0.1, 0.6],
    Line[{{-1, Mono1Median}, {1, Mono1Median}}]},
  PlotStyle → Directive[GrayLevel[0.5], Opacity[0.7], AbsolutePointSize[3]],
  Frame → {{True, False}, {True, False}}, FrameStyle → Directive[Black, Thickness[Medium]],
  Axes → False, BaseStyle → {FontFamily → "Arial", FontSize → 12}, AspectRatio → 1,
  ImageSize → {{1000}, {200}},
  FrameTicks → {Table[{i, i, {0.02, 0}}, {i, -1, 1, 1 / 2}], Table[{i, i, {0.02, 0}}, {i, 0, 20, 3}]},
  FrameLabel → {"Correlation in response", "Median PFS"}]
]

Export[NotebookDirectory[] <> "Supplementary Figure S10, Median PFS vs Correlation.pdf",
 %, "PDF"]
```

Out[99]=