Importing trial data

USER INPUT - treatment names, and file names containing event data.

```
Therapy1Name = "Paclitaxel_Carboplatin";
Therapy1FileName = "Paclitaxel_Carboplatin_events.csv";

Therapy2Name = "Olaparib";
Therapy2FileName = "Olaparib_events.csv";


(* please include a header in the source data which specifies the time units
 (e.g. weeks, or months). The same time units must be used for both therapies *)
Header = Import[NotebookDirectory[] <> Therapy1FileName, "CSV"][[1, 1]]
```

```
Time (months)
```

importing event data

```
Therapy1EventFile = Import[NotebookDirectory[] <> Therapy1FileName, "CSV"][[2 ;;]];
Therapy2EventFile = Import[NotebookDirectory[] <> Therapy2FileName, "CSV"][[2 ;;]];

Therapy1LongestEvent = Max[Therapy1EventFile[[All, 1]]]
Therapy2LongestEvent = Max[Therapy2EventFile[[All, 1]]]
SharedTimeFrame = Min[{Therapy1LongestEvent, Therapy2LongestEvent}]
```

```
19.1

24.7

19.1
```

Plotting trial data

```
Therapy1Color = RGBColor[0.5, 0.7, 0];
Therapy2Color = RGBColor[0.8, 0.1, 0.6];

Therapy1Events = EventData[Therapy1EventFile⟦All, 1⟧, Therapy1EventFile⟦All, 2⟧];
Therapy2Events = EventData[Therapy2EventFile⟦All, 1⟧, Therapy2EventFile⟦All, 2⟧];

Therapy1SurvivalModelFit[t_] := SurvivalModelFit[Therapy1Events][t]
Therapy2SurvivalModelFit[t_] := SurvivalModelFit[Therapy2Events][t]

Therapy1CensoredEvents =
  Cases[Transpose[{Therapy1Events["CensoredData"], Therapy1Events["CensoringIndicators"]}],
   {_, 1}];
Therapy2CensoredEvents =
  Cases[Transpose[{Therapy2Events["CensoredData"], Therapy2Events["CensoringIndicators"]}],
   {_, 1}];

Therapy1CensorGraphics =
  Graphics[Join[{AbsoluteThickness[1], Darker[Therapy1Color, 0.3]},
    Table[Line[{{i, Therapy1SurvivalModelFit[i] - .04},
       {i, Therapy1SurvivalModelFit[i] + .04}}], {i, Therapy1CensoredEvents⟦All, 1⟧}]]];
Therapy2CensorGraphics =
  Graphics[Join[{AbsoluteThickness[1], Darker[Therapy2Color, 0.3]},
    Table[Line[{{i, Therapy2SurvivalModelFit[i] - .04},
       {i, Therapy2SurvivalModelFit[i] + .04}}], {i, Therapy2CensoredEvents⟦All, 1⟧}]]];
```

```
Therapy1Plot = Show[
  Plot[SurvivalFunction[EmpiricalDistribution[Therapy1Events]][x],
   {x, 0, Therapy1LongestEvent}, Exclusions → None,
   PlotRange → {{0, Therapy1LongestEvent * 1.05}, {0, 1}}, PlotPoints → 500,
   Frame → {{True, False}, {True, False}}, Axes → False,
   FrameLabel → {Header, "Probability of\nprogression free survival (%)"},
   PlotRangePadding → None, BaseStyle → {FontFamily → "Arial", FontSize → 12},
   FrameStyle → Directive[Thickness[Medium], Black], AspectRatio → 2 / 3,
   ImageSize → {{1000}, {270}}, ImagePadding → {{80, 10}, {60, 10}},
   FrameTicks → {{Table[{i, 100 * i, {0.015, 0}}, {i, 0, 1, 2 / 10}], None}, {Automatic, None}},
   PlotLabel → Therapy1Name, PlotStyle → Directive[AbsoluteThickness[1], Therapy1Color]]
  ,
  Therapy1CensorGraphics
 ]
```



```
Therapy2Plot = Show[
  Plot[SurvivalFunction[EmpiricalDistribution[Therapy2Events]][x],
   {x, 0, Therapy2LongestEvent}, Exclusions → None,
   PlotRange → {{0, Therapy2LongestEvent * 1.05}, {0, 1}}, PlotPoints → 500,
   Frame → {{True, False}, {True, False}}, Axes → False,
   FrameLabel → {Header, "Probability of\nprogression free survival (%)"},
   PlotRangePadding → None, BaseStyle → {FontFamily → "Arial", FontSize → 12},
   FrameStyle → Directive[Thickness[Medium], Black], AspectRatio → 2 / 3,
   ImageSize → {{1000}, {270}}, ImagePadding → {{80, 10}, {60, 10}},
   FrameTicks → {{Table[{i, 100 * i, {0.015, 0}}, {i, 0, 1, 2 / 10}], None}, {Automatic, None}},
   PlotLabel → Therapy2Name, PlotStyle → Directive[AbsoluteThickness[1], Therapy2Color]]
  ,
  Therapy2CensorGraphics
 ]
```



Predicting effect of combination therapy with independent drug action

First, simulating with intermediate response correlation (approximately ρ = 0.3)

```
SampleSize = 10 000;
RandomSamplesFromTherapy1Response =
  Sort[RandomVariate[EmpiricalDistribution[Therapy1Events], SampleSize]];
RandomSamplesFromTherapy2Response =
  Sort[RandomVariate[EmpiricalDistribution[Therapy2Events], SampleSize]];

(* USER-ADJUSTABLE PARAMETER*)
IntermediateRankRandomization = 8500;
```

```
PartiallyRandomizedTherapy1Distribution =
  Sort[
   Table[{i + RandomReal[{-IntermediateRankRandomization, IntermediateRankRandomization}],
     RandomSamplesFromTherapy1Response[[i]]}, {i, 1, SampleSize}], #1[[1]] < #2[[1]] &];
PartiallyRandomizedTherapy2Distribution =
  Sort[
   Table[{i + RandomReal[{-IntermediateRankRandomization, IntermediateRankRandomization}],
     RandomSamplesFromTherapy2Response[[i]]}, {i, 1, SampleSize}], #1[[1]] < #2[[1]] &];

ResponseDistributionWithIntermediateCorrelation =
  Table[Max[{PartiallyRandomizedTherapy1Distribution[[i, 2]],
     PartiallyRandomizedTherapy2Distribution[[i, 2]]}], {i, 1, SampleSize}];
Print["Response correlation = " <>
  ToString[Round[SpearmanRho[PartiallyRandomizedTherapy1Distribution[[All, 2]],
     PartiallyRandomizedTherapy2Distribution[[All, 2]]], 0.001]]]
```

```
Response correlation = 0.305
```

If response correlation is higher or lower than the target value; decrease or increase (respectively) the value of ' IntermediateRankRandomization' above.

```
Table[
  RandomSamplesFromTherapy1Response =
   Sort[RandomVariate[EmpiricalDistribution[Therapy1Events], SampleSize]];
  RandomSamplesFromTherapy2Response =
   Sort[RandomVariate[EmpiricalDistribution[Therapy2Events], SampleSize]];
  PartiallyRandomizedTherapy1Distribution =
   Sort[
    Table[{i + RandomReal[{-IntermediateRankRandomization, IntermediateRankRandomization}],
      RandomSamplesFromTherapy1Response[[i]]},
     {i, 1, Length[RandomSamplesFromTherapy1Response]}], #1[[1]] < #2[[1]] &];
  PartiallyRandomizedTherapy2Distribution =
   Sort[
    Table[{i + RandomReal[{-IntermediateRankRandomization, IntermediateRankRandomization}],
      RandomSamplesFromTherapy2Response[[i]]},
     {i, 1, Length[RandomSamplesFromTherapy2Response]}], #1[[1]] < #2[[1]] &];

  SpearmanRho[PartiallyRandomizedTherapy1Distribution[[All, 2]],
   PartiallyRandomizedTherapy2Distribution[[All, 2]]]
  , {20}];

Print["Average response correlation = " <> ToString[Mean[%]]]
Histogram[%%, {0., 1.0, 0.02}, "Probability",
 PlotLabel → "Distribution of ρ across replicates"]
```

```
Average response correlation = 0.294293
```



Second, simulating with high response correlation (approximately ρ = 0.5)

```
SampleSize = 10000;
RandomSamplesFromTherapy1Response =
  Sort[RandomVariate[EmpiricalDistribution[Therapy1Events], SampleSize]];
RandomSamplesFromTherapy2Response =
  Sort[RandomVariate[EmpiricalDistribution[Therapy2Events], SampleSize]];

(* USER-ADJUSTABLE PARAMETER*)
HighRankRandomization = 5000;
```

```
PartiallyRandomizedTherapy1Distribution =
  Sort[Table[{i + RandomReal[{-HighRankRandomization, HighRankRandomization}],
     RandomSamplesFromTherapy1Response[[i]]}, {i, 1, SampleSize}], #1[[1]] < #2[[1]] &];
PartiallyRandomizedTherapy2Distribution =
  Sort[Table[{i + RandomReal[{-HighRankRandomization, HighRankRandomization}],
     RandomSamplesFromTherapy2Response[[i]]}, {i, 1, SampleSize}], #1[[1]] < #2[[1]] &];

ResponseDistributionWithHighCorrelation =
  Table[Max[{PartiallyRandomizedTherapy1Distribution[[i, 2]],
     PartiallyRandomizedTherapy2Distribution[[i, 2]]}], {i, 1, SampleSize}];
Print["Response correlation = " <>
  ToString[Round[SpearmanRho[PartiallyRandomizedTherapy1Distribution[[All, 2]],
     PartiallyRandomizedTherapy2Distribution[[All, 2]]], 0.001]]]
```

```
Response correlation = 0.494
```

If response correlation is higher or lower than the target value; decrease or increase (respectively) the value of 'HighRankRandomization' above.

```
Table[
  RandomSamplesFromTherapy1Response =
   Sort[RandomVariate[EmpiricalDistribution[Therapy1Events], SampleSize]];
  RandomSamplesFromTherapy2Response =
   Sort[RandomVariate[EmpiricalDistribution[Therapy2Events], SampleSize]];
  PartiallyRandomizedTherapy1Distribution =
   Sort[Table[{i + RandomReal[{-HighRankRandomization, HighRankRandomization}],
      RandomSamplesFromTherapy1Response[[i]]},
     {i, 1, Length[RandomSamplesFromTherapy1Response]}], #1[[1]] < #2[[1]] &];
  PartiallyRandomizedTherapy2Distribution =
   Sort[Table[{i + RandomReal[{-HighRankRandomization, HighRankRandomization}],
      RandomSamplesFromTherapy2Response[[i]]},
     {i, 1, Length[RandomSamplesFromTherapy2Response]}], #1[[1]] < #2[[1]] &];

  SpearmanRho[PartiallyRandomizedTherapy1Distribution[[All, 2]],
   PartiallyRandomizedTherapy2Distribution[[All, 2]]]
  , {20}];

Print["Average response correlation = " <> ToString[Mean[%]]]
Histogram[%%, {0., 1.0, 0.02}, "Probability",
 PlotLabel → "Distribution of ρ across replicates"]
```

```
Average response correlation = 0.500721
```



Third, simulating with low response correlation (approximately ρ = 0.1)

```
SampleSize = 10000;
RandomSamplesFromTherapy1Response =
  Sort[RandomVariate[EmpiricalDistribution[Therapy1Events], SampleSize]];
RandomSamplesFromTherapy2Response =
  Sort[RandomVariate[EmpiricalDistribution[Therapy2Events], SampleSize]];

(* USER-ADJUSTABLE PARAMETER*)
LowRankRandomization = 25000;

PartiallyRandomizedTherapy1Distribution =
  Sort[Table[{i + RandomReal[{-LowRankRandomization, LowRankRandomization}],
     RandomSamplesFromTherapy1Response[[i]]}, {i, 1, SampleSize}], #1[[1]] < #2[[1]] &];
PartiallyRandomizedTherapy2Distribution =
  Sort[Table[{i + RandomReal[{-LowRankRandomization, LowRankRandomization}],
     RandomSamplesFromTherapy2Response[[i]]}, {i, 1, SampleSize}], #1[[1]] < #2[[1]] &];

ResponseDistributionWithLowCorrelation =
  Table[Max[{PartiallyRandomizedTherapy1Distribution[[i, 2]],
     PartiallyRandomizedTherapy2Distribution[[i, 2]]}], {i, 1, SampleSize}];
Print["Response correlation = " <>
  ToString[Round[SpearmanRho[PartiallyRandomizedTherapy1Distribution[[All, 2]],
     PartiallyRandomizedTherapy2Distribution[[All, 2]]], 0.001]]]
```

```
Response correlation = 0.111
```

If response correlation is higher or lower than the target value; decrease or increase (respectively) the value of 'LowRankRandomization' above.

```
Table[
  RandomSamplesFromTherapy1Response =
   Sort[RandomVariate[EmpiricalDistribution[Therapy1Events], SampleSize]];
  RandomSamplesFromTherapy2Response =
   Sort[RandomVariate[EmpiricalDistribution[Therapy2Events], SampleSize]];
  PartiallyRandomizedTherapy1Distribution =
   Sort[Table[{i + RandomReal[{-LowRankRandomization, LowRankRandomization}],
       RandomSamplesFromTherapy1Response[[i]]},
     {i, 1, Length[RandomSamplesFromTherapy1Response]}], #1[[1]] < #2[[1]] &];
  PartiallyRandomizedTherapy2Distribution =
   Sort[Table[{i + RandomReal[{-LowRankRandomization, LowRankRandomization}],
       RandomSamplesFromTherapy2Response[[i]]},
     {i, 1, Length[RandomSamplesFromTherapy2Response]}], #1[[1]] < #2[[1]] &];

  SpearmanRho[PartiallyRandomizedTherapy1Distribution[[All, 2]],
   PartiallyRandomizedTherapy2Distribution[[All, 2]]]
  , {20}];

Print["Average response correlation = " <> ToString[Mean[%]]]
Histogram[%%, {0., 1.0, 0.02}, "Probability",
 PlotLabel → "Distribution of ρ across replicates"]
```

```
Average response correlation = 0.0981574
```



Plotting prediction

```
Therapy1Median = Median[RandomVariate[EmpiricalDistribution[Therapy1Events], SampleSize]]
Therapy2Median = Median[RandomVariate[EmpiricalDistribution[Therapy2Events], SampleSize]]
Sim1Median = Median[ResponseDistributionWithLowCorrelation]
Sim2Median = Median[ResponseDistributionWithIntermediateCorrelation]
Sim3Median = Median[ResponseDistributionWithHighCorrelation]
```

```
9.7

9.

12.7

12.5

11.2
```

```
(* the following "offset" parameters adjust the position of the legend so that
 it can fit in a convenient portion of the plot *)
(*legend vertical offset*)
lvo = -0.1;
(* legend horizontal offset *)
lho = -0.5;

PredictionPlot = Show[
  Plot[{
    SurvivalFunction[EmpiricalDistribution[ResponseDistributionWithHighCorrelation]][x],
    SurvivalFunction[EmpiricalDistribution[
       ResponseDistributionWithIntermediateCorrelation]][x],
    SurvivalFunction[EmpiricalDistribution[ResponseDistributionWithLowCorrelation]][x],
    SurvivalFunction[EmpiricalDistribution[Therapy1Events]][x],
    SurvivalFunction[EmpiricalDistribution[Therapy2Events]][x]
   }
   , {x, 0, SharedTimeFrame}, PlotRange → {{0, Ceiling[SharedTimeFrame]}, {0, 1.11}},
   Exclusions → None,
   PlotStyle → {Directive[Black, Opacity[0]], Directive[Black, AbsoluteThickness[1.7]],
     Directive[Black, Opacity[0]], Directive[Therapy1Color, AbsoluteThickness[3]],
     Directive[Therapy2Color, AbsoluteThickness[3]]},
   BaseStyle → {FontFamily → "Arial", FontSize → 12},
   FrameStyle → Directive[Black, Thickness[Medium]], Frame → {{True, False}, {True, False}},
   FrameTicks → {{Table[{i, 100 * i, {0, 0.015}}, {i, 0, 1, 1 / 5}], None}, {Automatic, None}},
   FrameLabel → {Style[Header], Style["Probability of\nprogression free survival (%)"]},
   Prolog → {Gray, Thickness[Medium], Lighter[Yellow, 0.75], EdgeForm[None],
     Rectangle[{0, 1}, {Ceiling[SharedTimeFrame], 1.11}],
     Black, AbsoluteThickness[1.7], GrayLevel[0.65],
     Rectangle[{Sim1Median, 1}, {Sim3Median, 1.03}], Black, Opacity[1],
     Line[{{Sim2Median, 1}, {Sim2Median, 1.03}}], Text["Median:", {0.2, 1.03}, {-1, -1}],
     Darker[Therapy1Color, 0.1], Text[ToString[NumberForm[Therapy1Median, {3, 1}]],
      {Therapy1Median, 1.03}, {0, -1}], Darker[Therapy2Color, 0.1],
     Text[ToString[NumberForm[Therapy2Median, {3, 1}]], {Therapy2Median, 1.03}, {0, -1}],
     Black, Text[ToString[NumberForm[Sim2Median, {3, 1}]], {Sim2Median, 1.0}, {0, 1}],
     AbsoluteThickness[2], Darker[Therapy1Color, 0.1],
     Line[{{Therapy1Median, 1}, {Therapy1Median, 1.03}}], Darker[Therapy2Color, 0.1],
     Line[{{Therapy2Median, 1}, {Therapy2Median, 1.03}}]}, AspectRatio → 1 / 2,
   Filling → {1 → {3}}, FillingStyle -> Directive[GrayLevel[0.65], Opacity[1]],
```

```
  ImageSize → {{1000}, {250}}, ImagePadding → {{80, 10}, {60, 10}},
  PlotPoints → 500,
  Epilog -> {EdgeForm[Directive[Black, Thickness[Medium]]], Opacity[1],
    EdgeForm[None], CapForm["Butt"], AbsoluteThickness[3], Therapy1Color,
    Line[{{lho + 0.8, 0.34 + lvo}, {lho + 1.6, 0.34 + lvo}}], Therapy2Color,
    Line[{{lho + 0.8, 0.26 + lvo}, {lho + 1.6, 0.26 + lvo}}], GrayLevel[0.65],
    Rectangle[{lho + 0.8, 0.18 - 0.03 + lvo}, {lho + 1.6, 0.18 + 0.03 + lvo}], Black,
    Opacity[1], AbsoluteThickness[1.7],
    Line[{{lho + 0.8, 0.18 + lvo}, {lho + 1.6, 0.18 + lvo}}], FontFamily → "Arial",
    FontSize → 11, Text[Style[Therapy1Name, FontSize → 11], {lho + 1.8, 0.34 + lvo}, {-1, 0}],
    Text[Style[Therapy2Name, FontSize → 11], {lho + 1.8, 0.26 + lvo}, {-1, 0}],
    Text[Style["Simulated combination", FontSize → 11], {lho + 1.8, 0.18 + lvo}, {-1, 0}]}]
 ,
 Therapy1CensorGraphics
 ,
 Therapy2CensorGraphics
]
```




```
(* Exporting plot in rasterized format (PNG) *)
Export[NotebookDirectory[] <> Therapy1Name <> "_plus_" <> Therapy2Name <>
   "_predicted PFS.png",
  Magnify[PredictionPlot, 300(* resolution in DPI; currently set to 300 *) / 72], "PNG"];

(* Exporting plot in vector format (PDF) *)
Export[NotebookDirectory[] <> Therapy1Name <> "_plus_" <> Therapy2Name <>
   "_predicted PFS.pdf", PredictionPlot, "PDF"];
```

Calculating predicted hazard ratio of combination vs constituent therapies alone

Note that the error range in the hazard ratio depends on the population size in a hypothetical "trial" of the combination, and the duration on trial, that is, the rate of censoring events.

Here, as a simple demonstration, we compute error ranges assuming 100 patients receiving a combination therapy, with any that are progression-free after 12 months being ‘censored’. These parameters are adjustable.

```
TrialPopulationSize = 100 (* patients *);
CensoringTime = 12(* months *);

GenerateCensoredEventData[PatientResponses_, CensoringTime_] :=
 Module[{ResponsesShorterThanCensoringTime, ResponsesLongerThanCensoringTime},
  ResponsesShorterThanCensoringTime = Select[PatientResponses, # ≤ CensoringTime &];
  ResponsesLongerThanCensoringTime = Select[PatientResponses, # > CensoringTime &];
  EventData[Join[ResponsesShorterThanCensoringTime, ResponsesLongerThanCensoringTime],
   Join[Table[0, {Length[ResponsesShorterThanCensoringTime]}],
    Table[1, {Length[ResponsesLongerThanCensoringTime]}]]]
 ]

(* event data for simulated combinations are here established not as a fixed data set,
but a generating function so that we can characterize random variation between
 replicate trials
 (specifically, we aim to average over the results of very many such trials) *)
EventDataWithLowCorrelation[] :=
  GenerateCensoredEventData[
   Sort[RandomSample[ResponseDistributionWithLowCorrelation, TrialPopulationSize]],
   CensoringTime];
EventDataWithIntermediateCorrelation[] :=
  GenerateCensoredEventData[
   Sort[RandomSample[ResponseDistributionWithIntermediateCorrelation, TrialPopulationSize]],
   CensoringTime];
EventDataWithHighCorrelation[] :=
  GenerateCensoredEventData[
   Sort[RandomSample[ResponseDistributionWithHighCorrelation, TrialPopulationSize]],
   CensoringTime];

(* custom function to join two sets of event data -
 this is necessary to implement the Cox Proportional Hazards model *)
JoinEventData[EventData1_, EventData2_] :=
 EventData[Join[EventData1[[2, 1]], EventData2[[2, 1]]],
  Join[EventData1[[2, 2]], EventData2[[2, 2]]]]
```

```
NumberOfReplicateTrials = 100;

(* this function returns the relative risk, and confidence interval, in the format:
  (95% lower confidence interval, median estimate, 95% upper confidence interval )
*)
RelativeRiskCalculation[descriptors_, eventdata_, PrintTable_
  (* set to 1 to print out the statistical table of Cox Model output *)] := Module[{},
  MyModelFit = CoxModelFit[{descriptors, eventdata}, {treatment}, {treatment},
    NominalVariables → treatment];

  If[PrintTable == 1, Print[MyModelFit["ParameterTable"]]];

  RelativeRisk = MyModelFit["RelativeRisk"][[1]];
  RelativeRiskLowerConfidenceInterval =
   MyModelFit["RelativeRiskConfidenceIntervals"][[1, 1]];
  RelativeRiskUpperConfidenceInterval =
   MyModelFit["RelativeRiskConfidenceIntervals"][[1, 2]];

  {RelativeRiskLowerConfidenceInterval, RelativeRisk, RelativeRiskUpperConfidenceInterval}
 ]
```

Comparison with Therapy 1

```
Descriptors = Join[Table["Therapy 1 only", {Length[Therapy1Events[[2, 1]]]}],
   Table["Therapy Combination", {Length[EventDataWithIntermediateCorrelation[][[2, 1]]]}]];
```

- assuming low correlation (ρ = 0.1)

```
RelativeRiskCalculation[Descriptors,
  JoinEventData[Therapy1Events, EventDataWithLowCorrelation[]], 1];
RRTherapy1vsLowCorrelation = (* average of many repeated trials *)
 Quiet@
  Mean[Table[RelativeRiskCalculation[Descriptors,
     JoinEventData[Therapy1Events, EventDataWithLowCorrelation[]], 0], {50}]]
```

	Estimate	Standard Error	Relative Risk	Wald-χ^2	DF	P-Value
treatment[Therapy Combination]	−1.34779	0.212794	0.259813	40.1169	1	2.39207×10^{-10}

```
{0.223155, 0.333637, 0.498881}
```

- assuming intermediate correlation (ρ = 0.3)

```
RelativeRiskCalculation[Descriptors,
  JoinEventData[Therapy1Events, EventDataWithIntermediateCorrelation[]], 1];
RRTherapy1vsIntermediateCorrelation = (* average of many repeated trials *)
 Quiet@
  Mean[Table[RelativeRiskCalculation[Descriptors,
     JoinEventData[Therapy1Events, EventDataWithIntermediateCorrelation[]], 0], {50}]]
```

	Estimate	Standard Error	Relative Risk	Wald-χ^2	DF	P-Value
treatment[Therapy Combination]	−0.869833	0.198388	0.419022	19.2238	1	0.0000116254

```
{0.259715, 0.384891, 0.57049}
```

- assuming high correlation (ρ = 0.5)

```
RelativeRiskCalculation[Descriptors,
  JoinEventData[Therapy1Events, EventDataWithHighCorrelation[]], 1];
RRTherapy1vsHighCorrelation = (* average of many repeated trials *)
 Quiet@
  Mean[Table[RelativeRiskCalculation[Descriptors,
     JoinEventData[Therapy1Events, EventDataWithHighCorrelation[]], 0], {50}]]
```

	Estimate	Standard Error	Relative Risk	Wald-χ^2	DF	P-Value
treatment[Therapy Combination]	−0.604974	0.189164	0.546089	10.2281	1	0.00138317

```
{0.27202, 0.402517, 0.595688}
```

Comparison with Therapy 2

```
Descriptors = Join[Table["Therapy 2 only", {Length[Therapy2Events[[2, 1]]]}],
   Table["Therapy Combination", {Length[EventDataWithIntermediateCorrelation[][[2, 1]]]}]];
```

- assuming low correlation (ρ = 0.1)

```
RelativeRiskCalculation[Descriptors,
  JoinEventData[Therapy2Events, EventDataWithLowCorrelation[]], 1];
RRTherapy2vsLowCorrelation = (* average of many repeated trials *)
 Quiet@
  Mean[Table[RelativeRiskCalculation[Descriptors,
     JoinEventData[Therapy2Events, EventDataWithLowCorrelation[]], 0], {50}]]
```

	Estimate	Standard Error	Relative Risk	Wald-χ^2	DF	P-Value
treatment[Therapy Combination]	−0.976677	0.244041	0.37656	16.0168	1	0.000062783

```
{0.267165, 0.427482, 0.684089}
```

- assuming intermediate correlation (ρ = 0.3)

```
RelativeRiskCalculation[Descriptors,
  JoinEventData[Therapy2Events, EventDataWithIntermediateCorrelation[]], 1];
RRTherapy2vsIntermediateCorrelation = (* average of many repeated trials *)
 Quiet@
  Mean[Table[RelativeRiskCalculation[Descriptors,
     JoinEventData[Therapy2Events, EventDataWithIntermediateCorrelation[]], 0], {50}]]
```

	Estimate	Standard Error	Relative Risk	Wald-χ^2	DF	P-Value
treatment[Therapy Combination]	−0.808042	0.239835	0.44573	11.3512	1	0.000753994

```
{0.293223, 0.467061, 0.744034}
```

- assuming high correlation (ρ = 0.5)

```
RelativeRiskCalculation[Descriptors,
  JoinEventData[Therapy2Events, EventDataWithHighCorrelation[]], 1];
RRTherapy2vsHighCorrelation = (* average of many repeated trials *)
 Quiet@
  Mean[Table[RelativeRiskCalculation[Descriptors,
     JoinEventData[Therapy2Events, EventDataWithHighCorrelation[]], 0], {50}]]
```

	Estimate	Standard Error	Relative Risk	Wald-χ^2	DF	P-Value
treatment[Therapy Combination]	−0.947677	0.244314	0.38764	15.046	1	0.00010492

```
{0.313522, 0.49766, 0.790002}
```

Plotting predicted hazard ratios

```
Needs["ErrorBarPlots`"]

(* custom logarithmic tick marks *)
(*label rotation *)lr = 0;
(* set to "90 Degree" to have hazard ratio labels rotated by 90 degrees *)
HazardRatioLogTicks =
  Join[{{Log[10, 1], Rotate["1.0", lr], {0, 0.1}}, {Log[10, 2.0], Rotate["2.0", lr], {0, 0.1}},
    {Log[10, 0.5], Rotate["0.5", lr], {0, 0.1}}, {Log[10, 0.2], Rotate["0.2", lr], {0, 0.1}},
    {Log[10, 0.1], Rotate["0.1", lr], {0, 0.1}}},
   Table[{Log[10, i], , {0, 0.04}}, {i, 1.1, 1.9, 0.1}],
   Table[{Log[10, i], , {0, 0.04}}, {i, 0.1, 0.9, 0.1}]];

(* custom function for 'ErrorListPlot' to generate data point with error bars
 on logarithmic scale *)
LogErrorBar[xcoordinate_, LowerMedianUpperDatum_] :=
 {{xcoordinate, Log[10, LowerMedianUpperDatum[[2]]]},
  ErrorBar[{Log[10, LowerMedianUpperDatum[[3]]] - Log[10, LowerMedianUpperDatum[[2]]],
    Log[10, LowerMedianUpperDatum[[1]]] - Log[10, LowerMedianUpperDatum[[3]]]}]}

HazardRatioPlot = ErrorListPlot[{
   {LogErrorBar[1, RRTherapy1vsLowCorrelation]},
   {LogErrorBar[2, RRTherapy1vsIntermediateCorrelation]},
   {LogErrorBar[3, RRTherapy1vsHighCorrelation]},
   {LogErrorBar[6, RRTherapy2vsLowCorrelation]},
   {LogErrorBar[7, RRTherapy2vsIntermediateCorrelation]},
   {LogErrorBar[8, RRTherapy2vsHighCorrelation]}
  }, Axes → False, Frame → {{True, False}, {True, False}},
  FrameStyle → Directive[Black, Thickness[Medium]],
  BaseStyle → {FontSize → 12, FontFamily → "Arial"},
  PlotRange → {{0, 9}, {Log[10, 0.1], Log[10, 2]}}, AspectRatio → 1.8,
  Prolog → {Therapy1Color, AbsoluteThickness[2], Line[{{0, 0}, {4.5, 0}}],
    Therapy2Color, Line[{{4.5, 0}, {9, 0}}]}, ImageSize → {{1000}, {330}},
  ImagePadding → {{80, 30}, {80, 20}},
  FrameTicks → {{HazardRatioLogTicks, None},
    {{{1, Rotate["0.1", π / 2], {0, 0}}, {2, Rotate["ρ = 0.3", π / 2], {0, 0}},
      {3, Rotate["0.5", π / 2], {0, 0}}, {6, Rotate["0.1", π / 2], {0, 0}},
      {7, Rotate["ρ = 0.3", π / 2], {0, 0}}, {8, Rotate["0.5", π / 2], {0, 0}}}, None}},
  PlotStyle → Join[Table[Directive[Black, AbsoluteThickness[2], AbsolutePointSize[6]], {3}],
    Table[Directive[Black, AbsoluteThickness[2], AbsolutePointSize[6]], {3}]],
  FrameLabel → {{"Hazard Ratio\n(combination vs individual therapies)", ""},
    {"vs. Therapy 1 ; vs. Therapy 2", ""}}]
```




```
(* Exporting plot in rasterized format (PNG) *)
Export[NotebookDirectory[] <> Therapy1Name <> "_plus_" <> Therapy2Name <>
   "_predicted Hazard ratio.png",
  Magnify[HazardRatioPlot, 300(* resolution in DPI; currently set to 300 *) / 72], "PNG"];

(* Exporting plot in vector format (PDF) *)
Export[NotebookDirectory[] <> Therapy1Name <> "_plus_" <> Therapy2Name <>
   "_predicted Hazard ratio.pdf", HazardRatioPlot, "PDF"];
```