importing distributions of tumor volume changes

```
NivolumabVolumeChange = Import[NotebookDirectory[] <> "LarkinNivolumabVolumeChange.csv", "CSV"][[
    All, 1]];
IpilimumabVolumeChange = Import[NotebookDirectory[] <> "LarkinIpilimumabVolumeChange.csv", "CSV"][[
    All, 1]];
CombinationVolumeChange =
  Import[NotebookDirectory[] <> "LarkinCombinationVolumeChange.csv", "CSV"][[All, 1]];

(* examining some response statistics for nivolumab *)
dataset = NivolumabVolumeChange;
Print["shrink >30%"]
Length[Select[dataset, # ≤ -0.3 &]] / Length[dataset] // N
Print["|change| <30%"]
Length[Select[dataset, And[# > -0.3, # < 0.3] &]] / Length[dataset] // N
Print["grow >30%"]
Length[Select[dataset, # >= 0.3 &]] / Length[dataset] // N
```

```
shrink >30%
```

```
0.5373
```

```
|change| <30%
```

```
0.261438
```

```
grow >30%
```

```
0.201262
```

```
(* examining some response statistics for ipilimumab *)
dataset = IpilimumabVolumeChange;
Print["shrink >30%"]
Length[Select[dataset, # ≤ -0.3 &]] / Length[dataset] // N
Print["|change| <30%"]
Length[Select[dataset, And[# > -0.3, # < 0.3] &]] / Length[dataset] // N
Print["grow >30%"]
Length[Select[dataset, # >= 0.3 &]] / Length[dataset] // N
```

```
shrink >30%
```

```
0.24191
```

```
|change| <30%
```

```
0.474639
```

```
grow >30%
```

```
0.283452
```

```
(* sorting responses *)
SortedNivolumabVolumeChange = Sort[NivolumabVolumeChange, #1 > #2 &];
SortedIpilimumabVolumeChange = Sort[IpilimumabVolumeChange, #1 > #2 &];
SortedCombinationVolumeChange = Sort[CombinationVolumeChange, #1 > #2 &];

(* assigning a horizontal coordinate to each point such that each distribution can
 be plotted on the same horizontal scale, ranging from 0 to 1 *)
HorizontallyNormalizedNivolumabVolumeChange =
  Table[{(i - 1) / (Length[NivolumabVolumeChange] - 1), SortedNivolumabVolumeChange[[i]]},
   {i, 1, Length[NivolumabVolumeChange]}];
HorizontallyNormalizedIpilimumabVolumeChange =
  Table[{(i - 1) / (Length[IpilimumabVolumeChange] - 1), SortedIpilimumabVolumeChange[[i]]},
   {i, 1, Length[IpilimumabVolumeChange]}];
HorizontallyNormalizedCombinationVolumeChange =
  Table[{(i - 1) / (Length[CombinationVolumeChange] - 1), SortedCombinationVolumeChange[[i]]},
   {i, 1, Length[CombinationVolumeChange]}];
```

simulation with random sampling ($\rho = 0$)

```
(* simulate 10,000 patients whose responses to Nivolumab and Ipilimumab are randomly
 drawn from those drugs' observe response distributions (as monotherapies),
with each patients response to the combination being the best one of the two
 monotherapy responses *)
BestOfRandomSamples =
  Sort[
   Table[Min[{RandomSample[NivolumabVolumeChange, 1], RandomSample[IpilimumabVolumeChange, 1]}],
    {10000}], #1 > #2 &];

(* assigning horizontal coordinates between 0 and 1 for plotting purposes *)
HorizontallyNormalizedRandomSamplesNormalizedValues =
  Table[{(i - 1) / (Length[BestOfRandomSamples] - 1), BestOfRandomSamples[[i]]},
   {i, 1, Length[BestOfRandomSamples]}];

(* calculating the fractions of tumors with >30% tumor shrinkage *)
IpilimumabNonResponseRate =
  Select[HorizontallyNormalizedIpilimumabVolumeChange, #[[2]] > -0.3 &][[-1, 1]] // N;
IpilimumabResponseRate = 100 * Round[1 - IpilimumabNonResponseRate, 1 / 100]
NivolumabNonResponseRate =
  Select[HorizontallyNormalizedNivolumabVolumeChange, #[[2]] > -0.3 &][[-1, 1]] // N;
NivolumabResponseRate = 100 * Round[1 - NivolumabNonResponseRate, 1 / 100]
CombinationNonResponseRate =
  Select[HorizontallyNormalizedCombinationVolumeChange, #[[2]] > -0.3 &][[-1, 1]] // N;
CombinationResponseRate = 100 * Round[1 - CombinationNonResponseRate, 1 / 100]
RandomSamplingNonResponseRate =
  Select[HorizontallyNormalizedRandomSamplesNormalizedValues, #[[2]] > -0.3 &][[-1, 1]] // N;
RandomSamplingResponseRate = 100 * Round[1 - RandomSamplingNonResponseRate, 1 / 100]
```

```
24
```

```
54
```

```
65
```

```
65
```

```
(* this command necessary for plot legends to not interfere with proper image size
 of PDF exports *)
SetOptions[$FrontEndSession, PrintingStyleEnvironment → "Working"]

(* a plot with simple, random sampling of monotherapy responses *)
WaterFallPlot =
 ListPlot[{HorizontallyNormalizedIpilimumabVolumeChange,
   HorizontallyNormalizedNivolumabVolumeChange,
   HorizontallyNormalizedRandomSamplesNormalizedValues,
   HorizontallyNormalizedCombinationVolumeChange}, Joined → True, Filling → None
  (*Axis*), Axes → False, Frame → {{True, False}, {True, False}}, PlotRangePadding → None,
  PlotRange → {{0, 1}, {-1.05, 1.05}}, AspectRatio → 1 / 2,
  FrameStyle → Directive[Black, Thickness[Medium]],
  FrameTicks → {None, Table[{i, 100 * i, {0, 0.01}}, {i, -1, 1, 1 / 4}]},
  PlotStyle → {Directive[AbsoluteThickness[2], RGBColor[0.8, 0.1, 0.6]],
    Directive[AbsoluteThickness[2], ColorData[3, 4]], Directive[AbsoluteThickness[2], Black],
    Directive[AbsoluteThickness[2], ColorData[3, 6]]},
  BaseStyle → {FontFamily → "Arial", FontSize → 12},
  FrameLabel → {"Patients", "Best change from baseline \n tumor volume (%) "},
  PlotLegends → {"Ipilimumab", "Nivolumab", "Random sampling of\nmonotherapy responses",
    "Combination"}, ImageSize → {{1000}, {250}}, ImagePadding → {{90, 10}, {60, 10}},
  Epilog → {Black, Thickness[Medium], Line[{{0, 0}, {1, 0}}]},
  Prolog → {Gray, Dashing[{0.025, 0.025}], Directive[Thickness[Medium]],
    Line[{{0, -0.3}, {1, -0.3}}], Dashing[None], RGBColor[0.8, 0.1, 0.6],
    Line[{{IpilimumabNonResponseRate, -0.3}, {IpilimumabNonResponseRate + 0.015, -0.3 + 0.08}}],
    Text[ToString[IpilimumabResponseRate] <> "%", {IpilimumabNonResponseRate + 0.015, -0.3 + 0.06},
     {-1, -1}], ColorData[3, 4],
    Line[{{NivolumabNonResponseRate, -0.3}, {NivolumabNonResponseRate + 0.015, -0.3 + 0.08}}],
    Text[ToString[NivolumabResponseRate] <> "%", {NivolumabNonResponseRate + 0.015, -0.3 + 0.06},
     {-1, -1}], Black,
    Line[{{CombinationNonResponseRate, -0.3},
      {CombinationNonResponseRate + 0.015, -0.3 - 0.08 * 2}}],
    Text[ToString[RandomSamplingResponseRate] <> "%",
     {RandomSamplingNonResponseRate + 0.01, -0.3 - 0.08 * 2}, {-1, 1}], ColorData[3, 6],
    Line[{{CombinationNonResponseRate, -0.3}, {CombinationNonResponseRate - 0.015, -0.3 - 0.08}}],
    Text[ToString[CombinationResponseRate] <> "%", {CombinationNonResponseRate, -0.3 - 0.08},
     {1, 1}]}]
```



simulation with correlated sampling (aiming for ρ=0.25)

```
GeneratePartiallyCorrelatedDraw := Module[{},

  (* one monotherapy distribution is slightly longer than the other
   (simply meaning that the image was more pixels wide);
  so the longer one is randomly sub-
   sampled to produce a list of matching length. The number of pixels
    (data points in thedistribution) remains vastly in excess of the number of
    patients (~16 pixels wide per patient),
  and so there is no meaningful loss of patient data. *)
  SubSampledNivolumabValues =
   Reverse[Sort[RandomChoice[NivolumabVolumeChange, Length[IpilimumabVolumeChange]]]];

  (* the two distributions begin sorted side by side (perfectly rank correlated),
  and each row is given some degree of 'rank randomization' that moves its position
   randomly up or down to some degree. Therefore with increasing value of the '
   amountofrankrandomization' parameter,
  the correlation decreases from ρ=
   1 towards ρ=0. The size of the 'amountofrankrandomization' parameter to produce
     a desired level of correlation is determined by trial and error *)
  amountofrankrandomization = 4300;
  SlightlyRandomizedMono1Distribution =
   Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
      SubSampledNivolumabValues[[i]]}, {i, 1, Length[SubSampledNivolumabValues]}],
    #1[[1]] < #2[[1]] &];
  SlightlyRandomizedMono2Distribution =
   Sort[Table[{i + RandomReal[{-amountofrankrandomization, amountofrankrandomization}],
      IpilimumabVolumeChange[[i]]}, {i, 1, Length[IpilimumabVolumeChange]}], #1[[1]] < #2[[1]] &];

  (* each patient's response is the best one of their two randomly assigned responses *)
  BestOfSlightlyCorrelatedMonoPairs =
   Table[Min[{SlightlyRandomizedMono1Distribution[[i, 2]],
      SlightlyRandomizedMono2Distribution[[i, 2]]}],
    {i, 1, Length[SlightlyRandomizedMono1Distribution]}]
 ]
```

For a given amount of rank randomization, the exact correlation varies between repeats.

Here the process above is repeated 100 times to show that the rank correlation is tightly distributed around ρ = 0.25

```
RankCorrelations = Table[
    GeneratePartiallyCorrelatedDraw;
    SpearmanRho[SlightlyRandomizedMono1Distribution[[All, 2]],
     SlightlyRandomizedMono2Distribution[[All, 2]]]
    , {100}];

Median[RankCorrelations]
Mean[RankCorrelations]
```

```
0.253014
```

```
0.252221
```

```
Histogram[RankCorrelations, {0, 0.5, 0.01}]
```




```
GeneratePartiallyCorrelatedDraw;
SpearmanRho[SlightlyRandomizedMono1Distribution[[All, 2]],
 SlightlyRandomizedMono2Distribution[[All, 2]]]
SortedResponses = Reverse[Sort[BestOfSlightlyCorrelatedMonoPairs]];
HorizontallyNormalizedPartiallyCorrelatedSamples =
  Table[{(i - 1) / (Length[SortedResponses] - 1), SortedResponses[[i]]},
   {i, 1, Length[SortedResponses]}];
```

```
0.253612
```

```
CorrelatedSamplingNonResponseRate =
  Select[HorizontallyNormalizedPartiallyCorrelatedSamples, #[[2]] > -0.3 &][[-1, 1]] // N;
CorrelatedSamplingResponseRate = 100 * Round[1 - CorrelatedSamplingNonResponseRate, 1 / 100]
```

```
62
```

```
Figure1AWaterFallPlot =
 ListPlot[{{{0, 0}}, HorizontallyNormalizedIpilimumabVolumeChange,
   HorizontallyNormalizedNivolumabVolumeChange, HorizontallyNormalizedCombinationVolumeChange,
   {{0, 0}}, HorizontallyNormalizedRandomSamplesNormalizedValues,
   HorizontallyNormalizedPartiallyCorrelatedSamples}, Joined → True, Filling → None
  (*Axis*), Axes → False, Frame → {{True, False}, {True, False}}, PlotRangePadding → None,
  PlotRange → {{0, 1}, {-1.025, 1.025}}, AspectRatio → 1 / 2,
  FrameStyle → Directive[Black, Thickness[Medium]],
  FrameTicks → {None, Join[Table[{i, 100 * i, {0, 0.018}}, {i, -1, 1, 1 / 2}],
     Table[{i, , {0, 0.009}}, {i, -1, 1, 1 / 10}]]},
  PlotStyle → {Opacity[0], Directive[AbsoluteThickness[3], RGBColor[0.8, 0.1, 0.6]],
    Directive[AbsoluteThickness[3], ColorData[3, 4]],
    Directive[AbsoluteThickness[3], ColorData[3, 6]], Opacity[0],
    Directive[AbsoluteThickness[1.7], GrayLevel[0.5]], Directive[AbsoluteThickness[1.7], Black]},
  BaseStyle → {FontFamily → "Arial", FontSize → 12},
  FrameLabel → {"Patients", "Best change from baseline \n tumor volume (%) "},
  PlotLegends → {"Clinical data:", "Ipilimumab", "Nivolumab", "Combination",
    "Sampling of monotherapy responses:", "Uncorrelated (ρ=0)",
    "Partially correlated (ρ=0.25)"}, ImageSize → {{1000}, {250}},
  ImagePadding → {{90, 10}, {60, 10}}, Epilog → {Black, Thickness[Medium], Line[{{0, 0}, {1, 0}}]},
  Prolog → {Gray, Dashing[{0.025, 0.025}], Directive[Thickness[Medium]],
    Line[{{0, -0.3}, {1, -0.3}}], Dashing[None], RGBColor[0.8, 0.1, 0.6],
    Line[{{IpilimumabNonResponseRate, -0.3}, {IpilimumabNonResponseRate + 0.015, -0.3 + 0.08}}],
    Text[ToString[IpilimumabResponseRate] <> "%", {IpilimumabNonResponseRate + 0.015, -0.3 + 0.06},
     {-1, -1}], ColorData[3, 4],
    Line[{{NivolumabNonResponseRate, -0.3}, {NivolumabNonResponseRate + 0.015, -0.3 + 0.08}}],
    Text[ToString[NivolumabResponseRate] <> "%", {NivolumabNonResponseRate + 0.015, -0.3 + 0.06},
     {-1, -1}], GrayLevel[0.5],
    Line[{{CombinationNonResponseRate, -0.3},
      {CombinationNonResponseRate + 0.015, -0.3 - 0.08 * 2}}],
    Text[ToString[RandomSamplingResponseRate] <> "%",
     {RandomSamplingNonResponseRate + 0.01, -0.3 - 0.08 * 2}, {-1, 1}], ColorData[3, 6],
    Line[{{CombinationNonResponseRate, -0.3}, {CombinationNonResponseRate - 0.015, -0.3 - 0.08}}],
    Text[ToString[CombinationResponseRate] <> "%", {CombinationNonResponseRate, -0.3 - 0.08},
     {1, 1}], Black, Text["numbers: response rate", {0.015, (-0.825)}, {-1, -1}],
    Text["(by ≥30% tumor shrinkage)", {0.015, -1}, {-1, -1}]}]

Export[NotebookDirectory[] <> "Figure 1A, tumor volume change.pdf", Figure1AWaterFallPlot,
 "PDF"]
```